

```

10 MODE 2
20 DEFINT I,K,N:DEFSTR F-H
30 PRINT "THIS PROGRAM PRODUCES A PLOT OF X AND Y VALUES"
40 PRINT "INPUT FROM THE KEYBOARD AND FINDS THE BEST-FIT"
50 PRINT "STRAIGHT LINE THROUGH THE GIVEN POINTS."
60 PRINT
70 PRINT "INPUT THE X AND Y CO-ORDINATES
      (SEPARATED BY A COMMA)"
80 PRINT "FOR EACH POINT."
90 PRINT
100 INPUT "HOW MANY POINTS WILL YOU INPUT";N
110 DIM X(N),Y(N)
120 PRINT
130 FOR I=1 TO N
140 PRINT "X(";I;"),Y(";I;") ";:INPUT X(I),Y(I)
150 PRINT
160 NEXT I
170 AX=X(1):CX=AX:AY=Y(1):CY=AY
180 FOR I=2 TO N
190 BX=X(I):BY=Y(I)
200 PX=MIN(AX,BX):QX=MAX(CX,BX)
210 PY=MIN(AY,BY):QY=MAX(CY,BY)
220 AX=PX:CY=QY:AY=PY:CY=QY
230 NEXT I
240 PRINT "THE X DATA SUPPLIED COVERS THE
      RANGE";PX;"TO";QX
250 PRINT "HOW DO YOU WISH TO SCALE THE X AXIS ?"
260 PRINT
270 INPUT "FROM A MINIMUM SCALE VALUE OF";X1
280 INPUT " TO A MAXIMUM SCALE VALUE OF";X2
290 INPUT " IN STEPS OF";X3
300 PRINT
310 PRINT
320 PRINT "THE Y DATA SUPPLIED COVERS THE RANGE"
      ;PY;"TO";QY
330 PRINT "HOW DO YOU WISH TO SCALE THE Y AXIS ?"
340 PRINT
350 INPUT "FROM A MINIMUM SCALE VALUE OF";Y1
360 INPUT " TO A MAXIMUM SCALE VALUE OF";Y2
370 INPUT " IN STEPS OF";Y3
380 PRINT
390 INPUT "WHAT LABEL FOR THE X AXIS";FX
400 INPUT "WHAT LABEL FOR THE Y AXIS";FY
410 INPUT " WHAT TITLE FOR THE PLOT";FT
420 PRINT
430 CLS
440 RX=X2-X1:RY=Y2-Y1:SX=300/RX:SY=300/RY
450 TAG
460 FOR I=1 TO N
470 KX=INT(SX*(X(I)-X1)):KY=INT(SY*(Y(I)-Y1))
480 IF KX>300 OR KY>300 THEN 550
490 IF TEST(KX+53,KY+55)=0 THEN 530
500 MOVE KX+50,KY+58
510 PRINT "*";
520 GOTO 550
530 MOVE KX+50,KY+58
540 PRINT "+";
550 NEXT I
560 NXT=RX/X3:SXT=SX*X3

```

```

570 FOR I=1 TO NXT
580 MOVE 50+(SXT*I),50
590 PRINT CHR$(145);
600 MOVE 42+(SXT*I),37
610 PRINT X1+(I*X3);
620 NEXT I
630 MOVE 50,50:PRINT CHR$(145);
640 MOVE 42,37:PRINT X1;
650 MOVE 350,50:PRINT CHR$(145);
660 MOVE 342,37:PRINT X2;
670 NYT=RY/Y3:SYT=SY*Y3
680 FOR I=1 TO NYT
690 MOVE 44,(SYT*I)+58
700 PRINT "-";
710 MOVE 8,(SYT*I)+58
720 PRINT Y1+(I*Y3);
730 NEXT I
740 MOVE 44,58:PRINT "-";
750 MOVE 8,58:PRINT Y1;
760 MOVE 44,358:PRINT "-";
770 MOVE 8,358:PRINT Y2;
780 MOVE 353,50:DRAW 50,50:DRAW 50,350
790 FOR I=50 TO 354 STEP 4:PLOT I,354:NEXT I
800 FOR I=354 TO 50 STEP -4:PLOT 354,I:NEXT I
810 TAGOFF
820 LOCATE 1,2:PRINT FY
830 LOCATE 20,25:PRINT FX
840 LOCATE 20,1:PRINT FT
850 WINDOW #1, 55,80,8,25
860 WINDOW #2, 55,80,1,5:PAPER #2,1:PEN #2,0
870 PRINT #1, "DO YOU WISH TO RESCALE"
880 PRINT #1, "THE AXES OF THE GRAPH"
890 INPUT #1, "(Y OR N)";G
900 IF UPPER$(G)="Y" THEN CLS:GOTO 240
910 T1=0:T2=0:T3=0:T4=0:T5=0:SD=0:DY=0
920 FOR I=1 TO N
930 T1=T1+(X(I)*Y(I)):T2=T2+X(I):T3=T3+Y(I):T4=T4+
      (X(I)*X(I)):T5=T5+(Y(I)*Y(I))
940 NEXT I
950 MX=T2/N:MY=T3/N
960 FOR I=1 TO N:SD=SD+(X(I)-MX)^2:DY=DY+(Y(I)-MY)
      ^2:NEXT I
970 SR=((T1-(T2*T3/N))^2)/(T4-((T2*T2)/N))
980 ST=T5-(T3*T3/N):RS=ST-SR:MS=RS/(N-2)
990 SL=(T1-(T2*T3/N))/(T4-(T2*T2/N))
1000 YI=(T3/N)-(SL*MX):W=SQR(DY/SD):R=SL/W
1010 CLS #1:CLS #2
1020 PRINT #2,"REGRESSION EQUATION IS OF"
1030 PRINT #2,"THE FORM y = a + bx"
1040 PRINT #2,"WHERE a ="; USING "####.###"; YI
1050 PRINT #2," b ="; USING "####.###"; SL
1060 PRINT #2,"CORR COEFF ="; USING "####.###"; R
1070 Y0=(SL*X2)+YI:YV=(SL*X1)+YI
1080 IF (YV>=Y1 AND YV<=Y2) AND (Y0>=Y1 AND
      Y0<=Y2) THEN 1150
1090 IF (YV>=Y1 AND YV<=Y2) AND (Y0>Y2) THEN 1180
1100 IF (YV<Y1) AND (Y0>=Y1 AND Y0<=Y2) THEN 1210
1110 IF (YV<Y1) AND (Y0>Y2) THEN 1240
1120 IF (YV>=Y1 AND YV<=Y2) AND (Y0<Y1) THEN 1270

```

```
1130 IF (YV>Y2) AND (YD>=Y1 AND YD<=Y2) THEN 1300
1140 IF (YV>Y2) AND (YD<Y1) THEN 1330
1150 X1ORD=50:Y1ORD=INT(SY*(((SL*X1)+YI)-Y1))+50
1160 X2ORD=350:Y2ORD=INT(SY*(((SL*X2)+YI)-Y1))+50
1170 GOTO 1350
1180 X1ORD=50:Y1ORD=INT(SY*(((SL*X1)+YI)-Y1))+50
1190 X2ORD=INT(((Y2-YI)/SL)-X1)*SX)+50:Y2ORD=350
1200 GOTO 1350
1210 X1ORD=INT(((Y1-YI)/SL)-X1)*SX)+50:Y1ORD=50
1220 X2ORD=350:Y2ORD=INT(SY*(((SL*X2)+YI)-Y1))+50
1230 GOTO 1350
1240 X1ORD=INT(((Y1-YI)/SL)-X1)*SX)+50:Y1ORD=50
1250 X2ORD=INT(((Y2-YI)/SL)-X1)*SX)+50:Y2ORD=350
1260 GOTO 1350
1270 X1ORD=50:Y1ORD=INT(SY*(((SL*X1)+YI)-Y1))+50
1280 X2ORD=INT(((Y1-YI)/SL)-X1)*SX)+50:Y2ORD=50
1290 GOTO 1350
1300 X1ORD=INT(((Y2-YI)/SL)-X1)*SX)+50:Y1ORD=350
1310 X2ORD=350:Y2ORD=INT(SY*(((SL*X2)+YI)-Y1))+50
1320 GOTO 1350
1330 X1ORD=INT(((Y2-YI)/SL)-X1)*SX)+50:Y1ORD=350
1340 X2ORD=INT(((Y1-YI)/SL)-X1)*SX)+50:Y2ORD=50
1350 MOVE X1ORD+3,Y1ORD+3:DRAW X2ORD+3,Y2ORD+3
1360 PRINT #1
1370 PRINT #1,"DO YOU WISH TO USE THE"
1380 PRINT #1,"REGRESSION EQUATION TO"
```

```
1390 PRINT #1,"ESTIMATE A VALUE OF"
1400 INPUT #1,"X OR Y (Y OR N)";G
1410 IF UPPER$(G)="N" THEN 1600
1420 CLS#1
1430 PRINT #1,"DO YOU WISH TO ESTIMATE"
1440 INPUT #1,"X OR Y";GI:PRINT #1
1450 IF UPPER$(GI)<>"X" AND UPPER$(GI)<>"Y" THEN 1420
1460 IF UPPER$(GI)="X" THEN 1520
1470 PRINT #1,"FOR WHAT VALUE OF X"
1480 INPUT #1,XH
1490 YH=YI+SL*XH:PRINT #1
1500 PRINT #1,"Y IS ESTIMATED AS"
1510 PRINT #1,YH:GOTO 1570
1520 PRINT #1,"FOR WHAT VALUE OF Y"
1530 INPUT #1,YH
1540 XH=(YH-YI)/SL:PRINT #1
1550 PRINT #1,"X IS ESTIMATED AS"
1560 PRINT #1,XH
1570 PRINT #1:PRINT #1,"CALCULATE ANOTHER"
1580 INPUT #1,"POINT (Y OR N)";G
1590 IF UPPER$(G)="Y" THEN 1420
1600 PRINT #1:PRINT #1,"DO YOU WISH TO PLOT"
1610 PRINT #1,"ANOTHER SET OF POINTS"
1620 INPUT #1,"(Y OR N)";G
1630 IF UPPER$(G)="Y" THEN RUN
1640 WINDOW SWAP 1,0
```